ORNL/CSD/TM-84

UCC-ND

NUCLEAR
DIVISION

UNION
CARBIDE

MASTER

Diffusion from Solid Cylinders

C. W. Nestor, Jr.

OPERATED BY
UNION CARBIDE CORPORATION
FOR THE UNITED STATES
DEPARTMENT OF ENERGY

ORNL/CSD/TM-84

Contract No. W-7405 eng 26

COMPUTER SCIENCES DIVISION

DIFFUSION FROM SOLID CYLINDERS

C. W. Nestor, Jr

Date Published - January 1980

Sponsor: H. W. Godbee
Originator: C. W. Nestor, Jr.

NOTICE This document contains information of a preliminary nature. It is subject to revision or correction and therefore does not represent a final report.

UNION CARBIDE CORPORATION, NUCLEAR DIVISION
operating the
Oak Ridge Gaseous Diffusion Plant
Oak Ridge Y-12 Plant
Oak Ridge National Laboratory
Paducah Gaseous Diffusion Plant
for the
DEPARTMENT OF ENERGY

CONTENTS

LIST OF FIGURES

ACKNOWLEDGMENT

The author would like to thank H. W. Godbee of the ORNL Chemical Technology Division for bringing this problem to his attention ano for providing the necessary financial support.

Thanks also are due the secretarial staff of the Computing Applications Department for their assistance in the preparation of this report.

DIFFUSION FROM SOLID CYLINDERS

C. W. Nestor, Jr.

ABSTRACT

The problem considered in this report is the diffusion of material from a solid cylinder initially containing a uniform concentration and immersed in a well-stirred bath which maintains the external concentration at zero. The Fourier-Bessel series form of the fraction of the original material removed from the cylinder as a function of time converges very slowly for small time. We have obtained an alternate form which converges reasonably rapidly for small time and have also used the convergence acceleration method of P. Wynn to provide an efficient method for computation. Numerical examples and program listings are included.

ANALYSIS

The diffusion equation in cylindrical geometry is

$$D\left[\frac{\partial^2 C}{\partial r^2} + \frac{1}{r}\frac{\partial C}{\partial r} + \frac{\partial^2 C}{\partial z^2}\right] = \frac{\partial C}{\partial t}\ , \quad o \le r \le a,\ \ -\ell \le z \le \ell \qquad (1)$$

where

C = the concentration of material inside the cylinder,

D = the diffusion coefficient,

a = the radius of the cylinder, and

ℓ = its half-height.

For the problem to be considered in this report, the initial condition is

$$C(r,z,o) = C_o$$

(a uniform initial concentration) and the boundary conditions are

$$\left.\begin{array}{l} C(a,z,t) = o \\ C(r,\ \pm\ell,\ t) = o \end{array}\right\}\quad t > o \quad .$$

The amount of material leaving the cylinder per unit time is

$$\int_{-\ell}^{\ell}\left[-D\,\frac{\partial C}{\partial r}\right]_{a,z,t} 2\pi a dz + 2\int_{o}^{a}\left[-D\,\frac{\partial C}{\partial z}\right]_{r,\ell,t} 2\pi r dr\ ,$$

and the fraction of the initial material leaving the cylinder in time t is

$$f(t) = \frac{\displaystyle\int_{o}^{t}\left\{\int_{-\ell}^{\ell}\left[-D\,\frac{\partial C}{\partial r}\right]_{a,z,\tau} 2\pi a dz + 2\int_{o}^{a}\left[-D\,\frac{\partial C}{\partial r}\right]_{r,\ell,\tau} 2\pi r dr\right\} d\tau}{2\pi a^2 \ell C_o}\ .$$

Taking Laplace transforms[1] $\left[F(s) = \int_0^\infty e^{-st} f(t)dt\right]$, we obtain

$$D\left[\frac{\partial^2 \bar{C}}{\partial r^2} + \frac{1}{r}\frac{\partial^2 \bar{C}}{\partial z^2}\right] = s\bar{C} - C_o$$

and[2]

$$F(s) = \frac{\int_{-\ell}^{\ell}\left[-D\frac{\partial \bar{C}}{\partial r}\right]_{a,z} 2\pi a dz + 2\int_0^a\left[-D\frac{\partial \bar{C}}{\partial z}\right]_{r,\ell} 2\pi r dr}{2\pi a^2 \ell C_o s} .$$

We assume a separable form for $\bar{C}$:

$$\bar{C}(r,z) = \bar{G}(r)\,\bar{H}(z)$$

and we can satisfy the boundary conditions by taking

$$\bar{G}(a) = o,$$

$$\bar{H}(\pm\ell) = o.$$

If we substitute

$$\bar{H}(z) = \sum_{n=1}^{\infty} A_n \cos \beta_n z$$

with $\beta_n = \frac{(2n-1)\pi}{2\ell}$, to satisfy the boundary conditions on the ends, into the partial differential equation for $\bar{C}$ and use the orthogonality property

$$\int_{-\ell}^{\ell} \cos \beta_k z \cos \beta_n z \, dz = \begin{cases} \ell, & k=n \\ o, & k\neq n \end{cases} ,$$

we obtain, for the coefficients A_n,

$$A_n\left[-D\left(\frac{d^2}{dr^2} + \frac{1}{r}\frac{d}{dr}\right)\bar{G}(r) + (s + \beta_n^2 D)\bar{G}(r)\right] = \frac{4C_o}{\pi}\,\frac{(-1)^{n-1}}{2n-1} .$$

The term in brackets must be independent of r. Let

$$\bar{G}(r) = \frac{1}{s+\beta_n^2 D} + \bar{Q}(r)$$

with

$$\bar{Q}(a) = \frac{-1}{s+\beta_n^2 D}$$

to satisfy the boundary condition at r = a; we have

$$\frac{d^2}{dr^2} + \frac{1}{r}\frac{d}{dr}\bar{Q}(r) - \mu^2 \bar{Q}(r) = 0$$

with

$$\mu^2 = \frac{s + D\beta_n^2}{D} \quad .$$

A solution of the differential equation for $\bar{Q}(r)$, satisfying the boundary condition at r = a, is

$$\bar{Q}(r) = -\frac{1}{s + D\beta_n^2}\frac{I_o(\mu r)}{I_o(\mu a)}$$

where $I_o(z)$ is the modified Bessel function of the first kind.[3] The Laplace transform of the concentration profile is

$$\bar{C}(r,z) = \frac{4C_o}{\pi}\sum_{n=1}^{\infty}\frac{(-1)^{n-1}}{2n-1}\frac{1}{s + D\beta_n^2}\left[1 - \frac{I_o(\mu r)}{I_o(\mu a)}\right]\cos\beta_n z$$

and, after inserting this into our expression for F(s), performing all the differentiations and integrations and collecting terms, we have

$$F(s) = \frac{2D}{\ell^2 s}\sum_{n=1}^{\infty}\frac{1}{s + D\beta_n^2} + \frac{16}{\pi^2}\sum_{n=1}^{\infty}\frac{1}{s + D\beta_n^2}\frac{I_1(\mu a)}{\mu a I_o(\mu a)}\frac{1}{(2n-1)^2} \quad .$$

Since[3]

$$I_o(z) = J_o(iz) \quad ,$$

the denominator of the Bessel function term will vanish for

$$\mu a = i\, j_{om}$$

where j_{om} is the m-th zero of $J_o(x)$, the Bessel function of the first kind. The singularities of F(s) are then all on the negative real axis:

$$s + D\beta_n^2 = -Dj_{om}^2/a^2$$

$$s = -D\,(\beta_n^2 + j_{om}^2/a^2)$$

and

$$s = -D\beta_n^2 \quad .$$

Inversion of F(s) by the method of residues[1] leads to the double series form for f(t):

$$f(t) = 1 - \frac{32}{\pi^2} \sum_{m=1}^{\infty} \frac{e^{-Dtj_{om}^2/a^2}}{j_{om}^2} \sum_{n=1}^{\infty} \frac{e^{-D\beta_n^2 t}}{(2n-1)^2} \quad .$$

For small time, this expression suffers from two serious computational difficulties. Both series converge very slowly for small time, so that a large number of terms must be included to give even modest accuracy; but also the result is very close to 1 and most of the significant figures are lost in the subtraction. An alternative expression can be obtained using the asymptotic expansions of the modified Bessel function[1,3]

$$I_o\,(z) \sim \frac{e^z}{\sqrt{2\pi z}} \left[1 + \frac{1}{8z} + \frac{9}{2(8z)^2} + \frac{75}{2(8z)^3} + \cdots\right]$$

$$I_1\,(z) \sim \frac{e^z}{\sqrt{2\pi z}} \left[1 - \frac{3}{8z} - \frac{15}{2(8z)^2} - \frac{105}{2(8z)^3} - \cdots\right]$$

$$\frac{I_1(z)}{I_0(z)} \sim 1 - \frac{1}{2z} - \frac{1}{8z^2} - \frac{1}{8z^3} - \cdots$$

$$\mu a = \frac{a}{\sqrt{D}} \sqrt{s+\beta_n^2 D}$$

$$F(s) = \sum_{n=1}^{\infty} \frac{1}{s+\beta_n^2 D} \left\{ \frac{2D}{\ell^2 s} \right.$$

$$\left. + \frac{16}{\pi^2}\left[\frac{\sqrt{D}}{a}\frac{1}{\sqrt{s+\beta_n^2 D}} - \frac{D}{2a^2}\frac{1}{s+\beta_n^2 D} - \frac{D^{3/2}}{8a^3}\frac{1}{(s+\beta_n^2 D)^{3/2}} - \frac{D^2}{8a^4}\frac{1}{(s+\beta_n^2 D)^2} - \cdots\right]\right\} .$$

Using the translation property

$$F(s+a) = L\left\{e^{-at} f(t)\right\} \quad ,$$

we can invert the series for F(s) term by term to give

$$f(t) = \frac{8}{\pi^2}\sum_{n=1}^{\infty} \frac{1-e^{-\beta_n^2 Dt}}{(2n-1)^2} +$$

$$\frac{16}{\pi^2}\left[\frac{D}{a}\,2\sqrt{\frac{t}{\pi}} - \frac{D}{2a^2}\,t - \frac{D^{3/2}}{8a^3}\,\frac{2t^{3/2}}{3\sqrt{\pi}} - \frac{D^2}{8a^4}\,\frac{t^2}{2} - \cdots\right]\sum_{n=1}^{\infty}\frac{e^{-\beta_n^2 Dt}}{(2n-1)^2}$$

$$f(t) \simeq \frac{8}{\pi^2}\sum_{n=1}^{\infty} \frac{1-e^{-\beta_n^2 Dt}}{(2n-1)^2} + \frac{8}{\pi^2}\left[\frac{4}{\sqrt{\pi}}\left(\frac{Dt}{a^2}\right)^{1/2} - \left(\frac{Dt}{a^2}\right) - \frac{1}{6\sqrt{\pi}}\left(\frac{Dt}{a^2}\right)^2\right]\sum_{n=1}^{\infty}\frac{e^{-\beta_n^2 Dt}}{(2n-1)^2} .$$

By a similar technique[1] we can obtain two mathematically equivalent forms for the finite slab problem and show that

$$\frac{8}{\pi^2}\sum_{n=1}^{\infty} \frac{1-e^{-\beta_n^2 Dt}}{(2n-1)^2} = \frac{2}{\sqrt{\pi}}\sqrt{\frac{Dt}{\ell^2}}\left[1 + 2\sqrt{\pi}\sum_{m=1}^{\infty}(-1)^m \operatorname{ierfc}\frac{m\ell}{\sqrt{Dt}}\right]$$

where[3]

$$\mathrm{ierfc}\,(x) = \int_x^\infty \mathrm{erfc}(t)dt$$

$$\mathrm{erfc}\,(x) = \frac{2}{\sqrt{\pi}}\int_x^\infty e^{-t^2}dt \quad .$$

The asymptotic expansion for the integrated complementary error function has a factor of e^{-z^2}, so that the sum of terms is negligible with respect to 1 for small time; we put

$$\frac{8}{\pi^2}\sum_{n=1}^{\infty}\frac{1-e^{-\beta_n^2 Dt}}{(2n-1)^2} \simeq \frac{2}{\sqrt{\pi}}\sqrt{\frac{Dt}{\ell^2}}$$

and

$$\frac{8}{\pi^2}\sum_{n=1}^{\infty}\frac{e^{-\beta_n^2 Dt}}{(2n-1)^2} \simeq 1 - \frac{2}{\sqrt{\pi}}\sqrt{\frac{Dt}{\ell^2}}$$

so that

$$f(t) \simeq \frac{2}{\sqrt{\pi}}\sqrt{\frac{Dt}{\ell^2}} + \left[1 - \frac{2}{\sqrt{\pi}}\frac{Dt}{\ell^2}\right]\left[\frac{4}{\sqrt{\pi}}\left(\frac{Dt}{a^2}\right)^{1/2} - \frac{Dt}{a^2} - \frac{1}{6\sqrt{\pi}}\left(\frac{Dt}{a^2}\right)^{3/2} - \frac{1}{8}\left(\frac{Dt}{a^2}\right)^2 - \dots\right] = \left(\frac{Dt}{a^2}\right)^{1/2}\left(\frac{4}{\sqrt{\pi}} + \frac{2}{\sqrt{\pi}}\frac{1}{\ell/a}\right) - \frac{Dt}{a^2}\left(1 + \frac{8}{\pi}\frac{1}{\ell/a}\right) - \left(\frac{Dt}{a^2}\right)^{3/2}\left(\frac{1}{6\sqrt{\pi}} - \frac{2}{\sqrt{\pi}}\frac{1}{\ell/a}\right) - \left(\frac{Dt}{a^2}\right)^2\left(\frac{1}{8} - \frac{1}{3\pi}\frac{1}{\ell/a}\right) - \dots \quad .$$

For small values of ℓ/a, a more useful dimensionless parameter is Dt/ℓ^2; our expression for f(t) becomes

$$f(t) = \left(\frac{Dt}{\ell^2}\right)^{1/2}\left[\frac{2}{\sqrt{\pi}} + \frac{4}{\sqrt{\pi}}\left(\frac{\ell}{a}\right)\right] - \frac{Dt}{\ell^2}\left[\frac{8}{\pi}\left(\frac{\ell}{a}\right) + \left(\frac{\ell}{a}\right)^2\right] + \left(\frac{Dt}{\ell^2}\right)^{3/2}\left[\frac{2}{\sqrt{\pi}}\left(\frac{\ell}{a}\right)^2 - \frac{1}{6\sqrt{\pi}}\left(\frac{\ell}{a}\right)^3\right] + \left(\frac{Dt}{\ell^2}\right)^2\left[\frac{1}{3\pi}\left(\frac{\ell}{a}\right)^3 - \frac{1}{8}\left(\frac{\ell}{a}\right)^4\right] + \dots$$

which reduces to the "infinite sheet" result for $\ell/a = 0$.

The surface of the cylinder is

$$S = 2\pi a^2 + 2\pi a(2\ell) \quad ;$$

its volume is

$$V = 2\pi a^2 \ell$$

so that the surface-to-volume ratio is

$$\frac{S}{V} = \frac{a^2+2a\ell}{a^2\ell} = \frac{1}{a}\left(2 + \frac{1}{\ell/a}\right) \quad .$$

The first term in the expansion for f(t) is

$$\left(\frac{Dt}{a^2}\right)^{1/2} \frac{2}{\sqrt{\pi}} \left(2 + \frac{1}{\ell/a}\right) = 2\,\frac{S}{V}\sqrt{\frac{Dt}{\pi}}$$

the semi-infinite medium result.

<u>Example</u>

The first four coefficients in the series expansion in $(Dt/a^2)^{1/2}$ are

$$C_1 = \frac{1}{\sqrt{\pi}} \quad (4 + \frac{2}{\ell/a})$$

$$C_2 = - \quad (1 + \frac{4}{\pi} \; \frac{2}{\ell/a})$$

$$C_3 = \frac{1}{\sqrt{\pi}} \quad (\frac{2}{\ell/a} - \frac{1}{6})$$

$$C_4 = \frac{1}{6\pi}\,\frac{2}{\ell/a} - \frac{1}{8} \quad .$$

For a cylinder of radius and half-height both 1 cm, the coefficients are

$$C_1 = \frac{6}{\sqrt{\pi}} = 3.38514$$

$$C_2 = -(1 + \frac{8}{\pi}) = -3.54648$$

$$C_3 = \frac{11}{6\sqrt{\pi}} = 1.03435$$

$$C_4 = \frac{1}{3\pi} - \frac{1}{8} = -0.018897 \quad .$$

The fraction leached as a function of the dimensionless parameter $\frac{Dt}{a^2}$ is shown in Table I.

Table I. Sample Calculation of f(t)

$\frac{Dt}{a^2}$	10^{-6}	10^{-4}	10^{-2}
f(t)	$3.382(10^{-3})$	$3.3498(10^{-2})$	0.30408

For $Dt/a^2 = 10^{-2}$, the eleventh term in the j_{om}^2 series is 9.73 (10^{-9}); the sum of the first eleven terms is 0.196132; the eleventh term in the $(2n-1)^2$ series is 4.05 (10^{-9}) and the sum is 1.094492; the series result for f(t) is then

$$f(t) = 0.303998$$

Our approximate value obtained from four terms agrees reasonably well. For smaller values of Dt/a^2, many more terms would be required in evaluating the series. If we require

$$e^{-\beta_n^2 Dt} < 10^{-8}$$

or

$$\beta_n^2 Dt > 16.1 ,$$

then, for $\frac{Dt}{a^2} = 10^{-4}$ and $\ell/a = 1$,

$$(2n-1)^2 \frac{\pi^2}{4} > 16.1(10^4)$$

$$n > 128$$

so of the order of 120 to 130 terms would be needed to attain reasonable accuracy in the series form.

CONVERGENCE ACCELERATION

For moderate values of time where the error after the first few terms of the expansion in $(Dt/a^2)^{1/2}$ would be unacceptably large, the convergence of the Fourier-Bessel series form can be considerably accelerated by the use of a scheme studied by (among others) P. Wynn.[5] The method is to determine a rational function in $1/n$ to match the partial sums of the series, and to extrapolate the rational function to $1/n = 0$. A portion of the calculation for the $(2n-1)^2$ series is shown in Table II, using a value of $Dt/\ell^2 = 10^{-4}$. We have

$$\frac{\ell}{\sqrt{Dt}} = 100$$

so the additional terms involving integrated complementary error functions are negligible, and the value of the sum, correct to six significant figures, is 1.21378. Extrapolating as far as possible after nine terms in the sum gives 1.21822. The successive columns of the table are generated by the "rhombus rule"

$$t_j^k = t_{j+1}^{k-2} + 1\Big/\left(t_{j+1}^{k-1} - t_j^{k-1}\right) \quad ,$$

and a convenient FORTRAN subroutine for generating successive upward-sloping diagonals of the table is given in the Appendix. (Subroutine EXTRAP.)

RESULTS

Sample calculations for cylinders of various sizes with a wide range of diffusion coefficients have been done with the computer program and subroutines listed in the Appendix. The results are shown in Figs. 1, 2, and 3 and tabulated in Table III.

Table II. A Portion of the Extrapolation Calculation

0.99975								
	9.02001							
1.11062		1.17260						
	25.1547		78.6592					
1.15037		1.19129		1.20444				
	49.5960		154.685		326.429			
1.17503		1.20080		1.21026		1.21456		
	82.6351		260.390		559.308		1025.80	
1.18264		1.20643		1.21361		1.21670		1.21822
	124.667		399.666		882.689		1683.26	
1.19066		1.21006		1.21658		1.21795		
	176.196		577.796		1323.00			
1.19633		1.21255		1.21702				
	237.844		801.721					
1.20054		1.21433						
	310.361							
1.20376								

Table III. Fraction Leached from Finite Cylinders

Dt/a^2	$\ell/a = 0.3$	$\ell/a = 0.5$	$\ell/a = 1.0$	$\ell/a = 3.0$	$\ell/a = 5.0$
.0003	.101408	.076360	.057574	.045045	.042545
.001	.180772	.136705	.103534	.081419	.076790
.003	.301748	.229279	.174942	.138664	.131416
.01	.510551	.392523	.303998	.244982	.233120
.02	.670959	.522430	.410497	.335875	.320950
.03	.771965	.610018	.484897	.401478	.384795
.04	.840159	.676009	.542837	.454011	.436245
.05	.887240	.728300	.590358	.498200	.479768
.06	.920116	.770865	.630548	.536468	.517652
.07	.943235	.806055	.665235	.570251	.551254
.08	.959574	.835428	.695608	.600484	.581459
.09	.971162	.860099	.722490	.627813	.608878
.10	.979401	.880912	.746475	.652708	.633954
.12	.989462	.913467	.787452	.696527	.678341
.15	.996127	.946170	.835666	.750653	.733647
.20	.999266	.975470	.891964	.818792	.804134
.25	.999861	.988796	.928648	.867662	.855401
.30	.999973	.994879	.952804	.903111	.893055
.40	.999999	.998930	.979326	.947847	.941336
.50	1.000000	.999776	.990941	.971829	.967746



Figure 1. Fraction Leached vs Dt/a^2 (log-log plot)



Figure 2. Fraction Leached vs Dt/a^2 (linear plot)



Figure 3. Fraction Leached vs Square Root of Dt/a^2

REFERENCES

1. H. S. Carslaw and J. C. Jaeger, *Operational Methods in Applied Mathematics*, Second Edition, Oxford, 1948.

2. G. E. Lee-Whiting, *One-Dimensional Diffusion with Time-Dependent Boundary Conditions and Diffusion Coefficient*, AECL-4374, January 1973.

3. M. Abramowitz and I. A. Stegun, eds., *Handbook of Mathematical Functions*, National Bureau of Standards, Applied Mathematics Series 55, June 1964, Chapt. 9.

4. H. W. Godbee and D. S. Joy, *Assessment of the Loss of Radioactive Isotopes from Waste Solids to the Environment. Part I: Background and Theory*, ORNL/TM-4333, February 1974.

5. See, for example, J. F. Hart et al., *Computer Approximations*, John Wiley and Sons, New York, 1968, p. 38.

APPENDIX

Computer Program and Subroutines

The computer program listed was used to generate the results shown in Table III and to produce the graphs shown in Figs. 1, 2, and 3. The subroutines called by the main program, except for FCALC and INCMP, are contained in the computer graphics package DISSPLA*. The total time required to compile, load, and execute the program was 1.2 minutes on the IBM 360/75; 7 seconds on the 360/91.

*A proprietary software product of Integrated Software Systems Corporation, San Diego, California.

```
C
C       FPLOT.FOR (29 JAN. 1979)
C
      REAL*8 DTOASQ(20), XLOA(5), FL(20,5)
      REAL*4 XP(20), YP(20)
      DATA DTOASQ /3.D-4, 1.D-3, 3.D-3, 1.D-2, 2.D-2, 3.D-2, 4.D-2,
     1 5.D-2, 6.D-2, 7.D-2, 8.D-2, .1D0, .12D0, .15D0, .2D0, .25D0,
     2  .3D0, .4D0, .5D0, 0.D0/
      DATA XLOA / 3D0,  5D0, 1.D0, 3.D0, 5.D0/
      DATA NDT /19/
      DO 95 N=1,NDT
        XP(N) = DTOASQ(N)
 95   CONTINUE
      CALL FCALC (DTOASQ, NDT, XLOA, 5, FL, 20)
      WRITE (6, 1) (XLOA(K), K=1,5)
 1    FORMAT (1H1, 20X, 'FRACTION LEACHED FROM FINITE CYLINDERS'/
     1 1H0, 18X, 5(3X, 'L/A =', F4.1, 2X)/
     2 1H0, 10X, 'DT/A**2'/1X)
      DO 97 N=1,NDT
        WRITE (6, 2) DTOASQ(N), (FL(N,K), K=1,5)
 2      FORMAT (10X, 1PE8.1, 5E14.5)
        IF (MOD (N, 5) .EQ. 0) WRITE (6, 3)
 3      FORMAT (1X)
 97   CONTINUE
      WRITE (6, 4)
 4    FORMAT (1H1)
      CALL INCMP
      CALL PSPLIN
      CALL SIMPLX
      CALL YAXANG (0.)
      CALL TITLE (0, 0, '*D&T/A!EH.6&2$', 100, '*F&RACTION *L&EACHED$',
     1 100, 10., 8.)
      CALL LOGLOG (1.E-4, 2.5, .01, 4.)
      CALL MSHIFT (0.1, -0.1)
      CALL RLMESS ('!A.5M5&L/!MXA-.5&A$', 100, 1.2E-4, .15)
      DO 105 K=1,4
        DO 100 N=1,NDT
          YP(N) = FL(N,K)
 100    CONTINUE
        CALL CURVE (XP, YP, 1, 1)
        CALL RLREAL (XLOA(K), 1, 1.E-4, YP(1))
        CALL CURVE (XP, YP, NDT, 0)
 105  CONTINUE
      CALL ENDPL (1)
```

```
      CALL RETITL
      CALL XTICKS (2)
      CALL GRAF (0., .1, .5, 0., .1, 1.)
      CALL MSHIFT (-0.6, 0.1)
      CALL RLMESS ('!A.5M5&L/!MXA-.5&A$', 100, .12, 1.05)
      DO 115 K=1,5
        IF (K .EQ. 5) CALL MSHIFT (-0.1, -0.3)
        MKNO = K + K - 2
        CALL MARKER (MKNO)
        DO 110 N=1,NDT
          YP(N) = FL(N,K)
  110   CONTINUE
        CALL CURVE (XP(13), YP(13), 1, 1)
        CALL RLREAL (XLOA(K), 1, XP(13), YP(13))
        CALL CURVE (XP, YP, NDT, 0)
  115 CONTINUE
      CALL RESET ('MSHIFT')
      CALL ENDPL (2)
      CALL TITLE (0, 0, '(*D&T/A!EH.6&2!EXHX&)!EH.6&1/2$', 100,
     1   '*F&RACTION *L&EACHED$', 100, 10., 8.)
      CALL XTICKS (2)
      CALL GRAF (0., .1, .5, 0., .1, 1.)
      NKP = 0
      DO 130 N=1,NDT
        TEST = DSQRT (DTOASQ(N))
        IF (TEST .GT. 0.5) GO TO 130
        NKP = NKP + 1
        XP(NKP) = TEST
C
  130 CONTINUE
      CALL MSHIFT (-0.5, 0.15)
      CALL RLMESS ('!A.5M5&L/!MXA-.5&A$', 100, XP(7) + .01, .95)
      DO 140 K=1,5
        MKNO = K + K - 2
        CALL MARKER (MKNO)
        DO 135 N=1,NDT
          YP(N) = FL(N,K)
  135   CONTINUE
        CALL CURVE (XP(7), YP(7), 1, 1)
        IF (K .EQ. 5) CALL MSHIFT (-0.3, -0.3)
        CALL RLREAL (XLOA(K), 1, XP(7), YP(7))
        CALL CURVE (XP, YP, NKP, 0)
  140 CONTINUE
      CALL ENDPL (3)
      CALL DONEPL
      STOP
      END
```

```
      SUBROUTINE FCALC (DTOASQ, NDT, XLOA, NXL, FRAC, IFD)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION DTOASQ(1), XLOA(1), FRAC(IFD, 1)
      DATA  RSRP /.564189583548D0/
      DATA FOROP /1.27323954474D0/, SMAX /1.D-4/
      DATA R6PI /5.3051647697D-2/
      DO 105 K=1,NXL
        TQ = 2.D0/XLOA(K)
        C1 = RSRP*(4.D0 + TQ)
        C2 = -1.D0 - FOROP*TQ
        C3 = RSRP*(TQ - .166666666667D0)
        C4 = R6PI*TQ - .125D0
        DO 100 N=1,NDT
          PSQ = DTOASQ(N)
          P = DSQRT (PSQ)
          SMT = DABS(C4*PSQ)
          IF (SMT .LE. SMAX) FRAC(N,K) = (((C4*P + C3)*P + C2)*P + C1)*P
          IF (SMT .GT. SMAX) CALL SERSUM (PSQ, XLOA(K), FRAC(N,K))
100     CONTINUE
105   CONTINUE
      RETURN
      END
      SUBROUTINE SERSUM (A, B, ANS)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION U(31), V(30)
      DATA CONST /3.242277876D0/, PIO2 /1.5707963268D0/
      EXTERNAL ATERM, BTERM
      CALL EXTRAP (A, ATERM, 15, 1.D-8, U, V, ASUM, INDA)
      IF (INDA .NE. 0) WRITE (6, 1)
1     FORMAT ('0A SERIES NOT CONVERGED')
      BT1 = A*(PIO2/B)**2
      CALL EXTRAP (BT1, BTERM, 15, 1.D-8, U, V, BSUM, INDB)
      IF (INDB .NE. 0) WRITE (6, 2)
2     FORMAT ('0B SERIES NOT CONVERGED')
      ANS = 1.D0 - CONST*ASUM*BSUM
      RETURN
      END
```

```
C
C        ATERM.FOR (30 AUGUST 1978)
C
      FUNCTION ATERM (N, A)
      IMPLICIT REAL*8 (A-H, O-Z)
      DIMENSION XJZT(20)
      DATA XJZT /
     1  2.4048255577D0,  5.5200781103D0,  8.6537279129D0,
     2 11.7915344391D0, 14.9309177086D0, 18.0710639679D0,
     3 21.2116366299D0, 24.3524715308D0, 27.4934791320D0,
     4 30.6346064684D0, 33.7758202136D0, 36.9170983537D0,
     5 40.0584257646D0, 43.1997917132D0, 46.3411883717D0,
     6 49.4826098974D0, 52.6240518411D0, 55.7655107550D0,
     7 58.9069839261D0, 62.0484691902D0/
      IF (N .LE. 20) XJZ = XJZT(N)
      IF (N .GT. 20) XJZ = XJZM(N)
      ANS = 0.D0
      ARG = A*(XJZ**2)
      IF (ARG .LT. 170.D0) ANS = DEXP(-ARG)/(XJZ**2)
      ATERM = ANS
      RETURN
      END
      FUNCTION XJZM(N)
      IMPLICIT REAL*8 (A-H, O-Z)
      DATA PI /3.14159265359D0/
      ATEB = PI*DFLOAT(8*N - 2)
      R8BT = 1.D0/ATEB
      T = R8BT**2
      XJZM = .125D0*ATEB + ((6046.4D0*T - 31.D0)*T*4.D0/3.D0 + 1.D0)
     1        *R8BT
      RETURN
      END
C
C        BTERM.FOR (30 AUGUST 1978)
C
      FUNCTION BTERM (N, B)
      IMPLICIT REAL*8 (A-H, O-Z)
      ODDSQ = DFLOAT ((N + N - 1)**2)
      ANS = 0.D0
      ARG = ODDSQ*B
      IF (ARG .LE. 170.D0) ANS = DEXP(-ARG)/ODDSQ
      BTERM = ANS
      RETURN
      END
```

```
C
C        EXTRAP.FOR (30 NOV. 1978)
C
      SUBROUTINE EXTRAP (T, TRAT, NMAX, EPS, U, V, ANS, IND)
C
C        P. WYNN'S EPSILON ALGORITHM (SEE HART ET AL.,
C        "COMPUTER APPROXIMATIONS", WILEY, NEW YORK, 1968, PG. 38.)
C
C
      IMPLICIT REAL*8 (A-H, O-Z)
      EXTERNAL TRAT
C
C        THE FUNCTION TRAT (N, T) COMPUTES THE N-TH TERM OF THE
C        SERIES TO BE SUMMED:
C
C         INF
C
C         SUM F (T)
C              N
C         N=1
C
      DIMENSION U(2), V(2)
C
C        THE ARRAYS U AND V MUST BE DIMENSIONED IN THE CALLING
C        PROGRAM WITH AT LEAST 2*NMAX AND 2*NMAX + 1
C        ELEMENTS, RESPECTIVELY. THEY WILL CONTAIN SUCCESSIVE
C        UPWARD-SLOPING DIAGONALS OF THE PADE' TABLE.
C
C        NMAX IS HALF THE MAXIMUM NUMBER OF TERMS,
C        ANS IS THE ESTIMATE OF THE SUM, AND
C        IND IS RETURNED AS ZERO IF THE EXTRAPOLATION CONVERGED,
C        WITH A RELATIVE ERROR OF EPS, BUT IS RETURNED AS -1
C        OTHERWISE.
C
      IND = 0
      TEMP = TRAT (1, T)
      U(1) = TEMP
      IF (DABS(TEMP) .LE. EPS) GO TO 120
      NV = 0
      NU = 1
      NN = 2
```

```
      DO 115 N=1,NMAX
        TNEW = TRAT (NN, T)
        V(1) = U(1) + TNEW
        TEMP = V(1)
        IF (DABS(TNEW/TEMP) .LT. EPS) GO TO 120
        V(2) = 1.D0/TNEW
        NN = NN + 1
        IF (NV .LT. 1) GO TO 105
        IQ = 1
        DO 100 KV=1,NV
          V(KV+2) = U(KV) + 1.D0/(V(KV+1) - U(KV+1))
          IF (IQ .EQ. 0) GO TO 100
          TEMP = V(KV+2)
          IF (DABS (V(KV)/TEMP - 1.D0) .LT. EPS) GO TO 120
C
 100    IQ = 1 - IQ
C
 105    TNEW = TRAT(NN, T)
        U(1) = V(1) + TNEW
        TEMP = U(1)
        IF (DABS(TNEW/TEMP) .LT. EPS) GO TO 120
        U(2) = 1.D0/TNEW
        NN = NN + 1
        IQ = 1
        DO 110 KU=1,NU
          U(KU+2) = V(KU) + 1.D0/(U(KU+1) - V(KU+1))
          IF (IQ .EQ. 0) GO TO 110
          TEMP = U(KU+2)
          IF (DABS (U(KU)/TEMP - 1.D0) .LT. EPS) GO TO 120
C
 110    IQ = 1 - IQ
        NV = NU + 1
        NU = NV + 1
C
 115  CONTINUE
      IND = -1
C
 120  ANS = TEMP
      RETURN
      END
```

```
C
C        INCMP.FOR (15 DEC. 1978)
C
      SUBROUTINE INCMP
C
C        INITIALIZES DISSPLA FOR 14 BY 11 INCH PLOTS IN
C        A COMPRESSED DATA SET
C
      CALL COMPRS
      CALL BGNPL (1)
      CALL PAGE (14., 11.)
      CALL HEIGHT (0.2)
      CALL MX1ALF ('L/CSTD', '&')
      CALL MX2ALF ('STANDA', '*')
      CALL MX3ALF ('INSTRU', '!')
      RETURN
      END
```